
Esempio

 Lista attività con totali per 
impiegato (EMPNO) e 
progetto (PROJNO)

Sorgenti esempio:
CICLO.RPGLE:
ciclo RPG tradizionale
CICLOSQL1.SQLRPGLE: 
simulazione ciclo RPG con SQL v. 1
CICLOSQL2.SQLRPGLE: 
simulazione ciclo RPG con SQL v. 2



Ciclo RPG/1

Fonte: IBM Programming IBM Rational Development Studio for i ILE RPG Reference

rotture di livello
LR

EMPNO         L2
PROJNO        L1

IPROJA         01
I                                          EMPNO         L2
I                                          PROJNO        L1



Ciclo RPG/2

 Rotture di livello - intestazione

 Ho in linea il primo record che ha causato la rottura di livello

 Inizializzazione variabili per totalizzazione

 Emissione informazioni di intestazione

C   L1 do
C                   clear                   Tot1
C                   eval Msg = '>>  L1 intestazione ' + EMPNO + '-' +
C                                   PROJNO
C                   snd-msg   Msg
C                   enddo



Ciclo RPG/3

 Rotture di livello - totali

 Ho in linea il record precedente a quello che causerà la rottura di livello
ovvero l'ultimo record del gruppo

 Incremento variabili totalizzazione livello successivo

 Emissione informazioni di riepilogo del livello

CL1 add Tot1 Tot2
CL1 eval Msg = '>>  L1 totali ' + EMPNO + '-' +
C                                   PROJNO +
C                                   ' ore: ' + %char(Tot1)
CL1 snd-msg   Msg

Come posso scrivere un 
programma RPG 

embedded SQL simulando 
la lettura sequenziale con 
rotture di livello tipica del 

ciclo RPG?



Funzioni SQL lag e lead

 lag e lead sono due funzioni disponibili nel gruppo delle OLAP-
specifications

 Le On-Line Analytical Processing specifications consentono di 
restituire informazioni calcolate tramite funzioni di aggregazione 
come valori scalari

 lag: restituisce il valore dell'n-simo record precedente a quello 
corrente

 lead: restituisce il valore dell'n-simo record successivo a quello 
corrente

https://www.ibm.com/docs/en/i/7.5?topic=expressions-olap-specifications

https://www.ibm.com/docs/en/i/7.5?topic=expressions-olap-specifications
https://www.ibm.com/docs/en/i/7.5?topic=expressions-olap-specifications


Simulare le rotture di livello

select case when lag(EMPNO) over(order by order of EMPPROJACT)
                  is distinct from EMPNO
            then '1' else '0' end "L2 int.",
       case when lead(EMPNO) over(order by order of EMPPROJACT)
                  is distinct from EMPNO
            then '1' else '0' end "L2 tot.",
       case when lag(PROJNO) over(partition by EMPNO order by order of EMPPROJACT)
                  is distinct from PROJNO
            then '1' else '0' end "L1 int.",
       case when lead(PROJNO) over(partition by EMPNO order by order of EMPPROJACT)
                  is distinct from PROJNO
            then '1' else '0' end "L1 tot.",
        EMPNO, PROJNO, ACTNO, EMSTDATE, EMPTIME
      from EMPPROJACT
      order by EMPNO, PROJNO, ACTNO;

record precedente in 
base a ordinamento

record successivo in 
base a ordinamento

record 
corrente

record 
corrente

simulazione indicatore 
Lx di testata

simulazione indicatore 
Lx di totale

L2
L1

dcl-ds oAct qualified inz;
L2h ind;
L2t ind;
L1h ind;
L1t ind;
EMPNO    like(EMPNO);
PROJNO   like(PROJNO);
ACTNO    like(ACTNO);
EMSTDATE like(EMSTDATE);
EMPTIME  like(EMPTIME);

end-ds;

ds output per 
fetch SQL

record successivo in base a 
ordinamento e partizionato in 

base al campo del livello 
superiore

record precedente in base a 
ordinamento e partizionato in 

base al campo del livello 
superiore



Simulare le rotture di livello: esempio

campi di dettaglioL2
L1

indicatori di livello

Il resultset finale:
- contiene gli stessi record di dettaglio come se 
si stesse leggendo il file con ciclo RPG
- per ogni record i primi 4 campi hanno gli stessi 
valori degli indicatori di livello di testata e 
totale del ciclo RPG

Le operazioni di totalizzazione ai vari livelli 
vanno gestite nel programma RPG!



Simulare le rotture di livello e totalizzare

indicatori di livello L2
L1 campi di dettaglio

Totale L2

Totale L1

Totale LR

Il resultset finale:
- contiene gli stessi record di dettaglio come se si stesse 
leggendo il file con ciclo RPG + i record con i subtotali per 
ogni livello + il totale generale
- per ogni record i primi 5 campi hanno gli stessi valori 
degli indicatori di livello di testata e totale del ciclo RPG

Il programma RPG NON deve calcolare i totali!



Simulare le rotture di livello e totalizzare

with

     TOT as
     (select EMPNO, PROJNO,
           sum(EMPTIME) as EMPTIME
      from EMPPROJACT
      group by EMPNO, PROJNO with rollup),

     DETT as
     (select EMPNO, PROJNO, ACTNO, EMSTDATE, EMPTIME
      from EMPPROJACT),

     RES as
     (select EMPNO, PROJNO,
             cast(null as integer) as ACTNO,
             cast(null as date) as EMSTDATE, EMPTIME
        from TOT
      union all
      select *
        from DETT)

. . . 

Totalizzazione di 
EMPTIME

Totali livello L1, L2 e 
LR

Dettaglio

Query di unione 
dettaglio + totali



Simulare le rotture di livello e totalizzare

. . . 
     select case when EMPNO is null and PROJNO is null then '1' else '0' end "LR tot.",
            case when lag(EMPNO)  over(order by order of RES)
                       is distinct from EMPNO and EMPNO is not null
                 then '1' else '0' end "L2 int.",
            case when lead(EMPNO) over(order by order of RES)
                       is distinct from EMPNO
                 then '1' else '0' end "L2 tot.",
            case when lag(PROJNO)  over(partition by EMPNO order by order of RES)
                       is distinct from PROJNO and PROJNO is not null
                 then '1' else '0' end "L1 int.",
            case when lead(PROJNO) over(partition by EMPNO order by order of RES)
                       is distinct from PROJNO
                 then '1' else '0' end "L1 tot.",
            RES.*
       from RES
       order by EMPNO, PROJNO, ACTNO

dcl-ds oAct qualified inz;
LRt ind;
L2h ind;
L2t ind;
L1h ind;
L1t ind;
EMPNO       like(EMPNO);
PROJNO      like(PROJNO);
ACTNO       like(ACTNO);
EMSTDATE    like(EMSTDATE);
TOT_EMPTIME like(EMPTIME);

end-ds;
dcl-s iAct int(5) dim(10);

ds output per fetch 
SQL e array indicator

variable

Aggiungo alla query di unione "totali + 
dettaglio" i campi per la simulazione 

indicatori di livello



Riferimenti

 E-mail aziendale: marco.riva@ivolution.it

 Blog: www.markonetools.it

 E-mail blog: info@markonetools.it

 Linkedin: www.linkedin.com/in/marcoriva-mk1

 Twitter: @MarcoRiva73

 Facebook: https://www.facebook.com/markonetools/

 YouTube: youtube.com/@markonetools

		INIZIO CICLO

		>   L2 intestazione 000010

		>>  L1 intestazione 000010-AD3100

		>>> dettaglio record 000010-AD3100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-AD3100 ore: 0		50

		>>  L1 intestazione 000010-MA2100

		>>> dettaglio record 000010-MA2100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-MA2100 ore: 0		50

		>>  L1 intestazione 000010-MA2110

		>>> dettaglio record 000010-MA2110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000010-MA2110 ore: 1		00

		>   L2 totali 000010 ore: 2		00

		>   L2 intestazione 000020

		>>  L1 intestazione 000020-PL2100

		>>> dettaglio record 000020-PL2100-30/1982-01-01 ore: 1		00

		>>  L1 totali 000020-PL2100 ore: 1		00

		>   L2 totali 000020 ore: 1		00

		>   L2 intestazione 000030

		>>  L1 intestazione 000030-IF1000

		>>> dettaglio record 000030-IF1000-10/1982-06-01 ore: 0		50

		>>  L1 totali 000030-IF1000 ore: 0		50

		>>  L1 intestazione 000030-IF2000

		>>> dettaglio record 000030-IF2000-10/1982-01-01 ore: 0		50

		>>  L1 totali 000030-IF2000 ore: 0		50

		>   L2 totali 000030 ore: 1		00

		>   L2 intestazione 000050

		>>  L1 intestazione 000050-OP1000

		>>> dettaglio record 000050-OP1000-10/1982-01-01 ore: 0		25

		>>  L1 totali 000050-OP1000 ore: 0		25

		>>  L1 intestazione 000050-OP2010

		>>> dettaglio record 000050-OP2010-10/1982-01-01 ore: 0		75

		>>  L1 totali 000050-OP2010 ore: 0		75

		>   L2 totali 000050 ore: 1		00

		>   L2 intestazione 000070

		>>  L1 intestazione 000070-AD3110

		>>> dettaglio record 000070-AD3110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000070-AD3110 ore: 1		00

		>   L2 totali 000070 ore: 1		00

		>   L2 intestazione 000090

		>>  L1 intestazione 000090-OP1010

		>>> dettaglio record 000090-OP1010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000090-OP1010 ore: 1		00

		>   L2 totali 000090 ore: 1		00

		>   L2 intestazione 000100

		>>  L1 intestazione 000100-OP2010

		>>> dettaglio record 000100-OP2010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000100-OP2010 ore: 1		00

		>   L2 totali 000100 ore: 1		00

		>   L2 intestazione 000110

		>>  L1 intestazione 000110-MA2100

		>>> dettaglio record 000110-MA2100-20/1982-01-01 ore: 1		00

		>>  L1 totali 000110-MA2100 ore: 1		00

		>   L2 totali 000110 ore: 1		00

		>   L2 intestazione 000130

		>>  L1 intestazione 000130-IF1000

		>>> dettaglio record 000130-IF1000-90/1982-10-01 ore: 1		00

		>>> dettaglio record 000130-IF1000-100/1982-10-01 ore: 0		50

		>>  L1 totali 000130-IF1000 ore: 1		50

		>   L2 totali 000130 ore: 1		50

		>   L2 intestazione 000140

		>>  L1 intestazione 000140-IF1000

		>>> dettaglio record 000140-IF1000-90/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF1000 ore: 0		50

		>>  L1 intestazione 000140-IF2000

		>>> dettaglio record 000140-IF2000-100/1982-01-01 ore: 1		00

		>>> dettaglio record 000140-IF2000-100/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF2000 ore: 2		50

		>   L2 totali 000140 ore: 3		00

		>   L2 intestazione 000150

		>>  L1 intestazione 000150-MA2112

		>>> dettaglio record 000150-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000150-MA2112-180/1982-07-15 ore: 1		00

		>>  L1 totali 000150-MA2112 ore: 2		00

		>   L2 totali 000150 ore: 2		00

		>   L2 intestazione 000160

		>>  L1 intestazione 000160-MA2113

		>>> dettaglio record 000160-MA2113-60/1982-07-15 ore: 1		00

		>>  L1 totali 000160-MA2113 ore: 1		00

		>   L2 totali 000160 ore: 1		00

		>   L2 intestazione 000170

		>>  L1 intestazione 000170-MA2112

		>>> dettaglio record 000170-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000170-MA2112-70/1982-06-01 ore: 1		00

		>>  L1 totali 000170-MA2112 ore: 2		00

		>>  L1 intestazione 000170-MA2113

		>>> dettaglio record 000170-MA2113-80/1982-01-01 ore: 1		00

		>>  L1 totali 000170-MA2113 ore: 1		00

		>   L2 totali 000170 ore: 3		00

		>   L2 intestazione 000180

		>>  L1 intestazione 000180-MA2113

		>>> dettaglio record 000180-MA2113-70/1982-04-01 ore: 1		00

		>>  L1 totali 000180-MA2113 ore: 1		00

		>   L2 totali 000180 ore: 1		00

		>   L2 intestazione 000190

		>>  L1 intestazione 000190-MA2112

		>>> dettaglio record 000190-MA2112-70/1982-01-01 ore: 1		00

		>>> dettaglio record 000190-MA2112-80/1982-10-01 ore: 1		00

		>>  L1 totali 000190-MA2112 ore: 2		00

		>   L2 totali 000190 ore: 2		00

		>   L2 intestazione 000200

		>>  L1 intestazione 000200-MA2111

		>>> dettaglio record 000200-MA2111-50/1982-01-01 ore: 1		00

		>>> dettaglio record 000200-MA2111-60/1982-06-15 ore: 1		00

		>>  L1 totali 000200-MA2111 ore: 2		00

		>   L2 totali 000200 ore: 2		00

		>   L2 intestazione 000210

		>>  L1 intestazione 000210-MA2113

		>>> dettaglio record 000210-MA2113-80/1982-10-01 ore: 0		50

		>>> dettaglio record 000210-MA2113-180/1982-10-01 ore: 0		50

		>>  L1 totali 000210-MA2113 ore: 1		00

		>   L2 totali 000210 ore: 1		00

		>   L2 intestazione 000220

		>>  L1 intestazione 000220-MA2111

		>>> dettaglio record 000220-MA2111-40/1982-01-01 ore: 1		00

		>>  L1 totali 000220-MA2111 ore: 1		00

		>   L2 totali 000220 ore: 1		00

		>   L2 intestazione 000230

		>>  L1 intestazione 000230-AD3111

		>>> dettaglio record 000230-AD3111-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000230-AD3111-60/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-70/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-80/1982-04-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-180/1982-10-15 ore: 0		50

		>>  L1 totali 000230-AD3111 ore: 3		00

		>   L2 totali 000230 ore: 3		00

		>   L2 intestazione 000240

		>>  L1 intestazione 000240-AD3111

		>>> dettaglio record 000240-AD3111-70/1982-02-15 ore: 1		00

		>>> dettaglio record 000240-AD3111-80/1982-09-15 ore: 1		00

		>>  L1 totali 000240-AD3111 ore: 2		00

		>   L2 totali 000240 ore: 2		00

		>   L2 intestazione 000250

		>>  L1 intestazione 000250-AD3112

		>>> dettaglio record 000250-AD3112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-60/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-60/1983-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-70/1982-03-15 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-10-15 ore: 0		50

		>>> dettaglio record 000250-AD3112-180/1982-08-15 ore: 0		50

		>>  L1 totali 000250-AD3112 ore: 5		50

		>   L2 totali 000250 ore: 5		50

		>   L2 intestazione 000260

		>>  L1 intestazione 000260-AD3113

		>>> dettaglio record 000260-AD3113-70/1982-06-15 ore: 0		50

		>>> dettaglio record 000260-AD3113-70/1982-07-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-04-15 ore: 1		00

		>>> dettaglio record 000260-AD3113-180/1982-06-01 ore: 1		00

		>>  L1 totali 000260-AD3113 ore: 5		50

		>   L2 totali 000260 ore: 5		50

		>   L2 intestazione 000270

		>>  L1 intestazione 000270-AD3113

		>>> dettaglio record 000270-AD3113-60/1982-03-01 ore: 0		50

		>>> dettaglio record 000270-AD3113-60/1982-04-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-60/1982-09-01 ore: 0		25

		>>> dettaglio record 000270-AD3113-70/1982-09-01 ore: 0		75

		>>> dettaglio record 000270-AD3113-70/1982-10-15 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-03-01 ore: 0		50

		>>  L1 totali 000270-AD3113 ore: 5		00

		>   L2 totali 000270 ore: 5		00

		>   L2 intestazione 000280

		>>  L1 intestazione 000280-OP1010

		>>> dettaglio record 000280-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000280-OP1010 ore: 1		00

		>   L2 totali 000280 ore: 1		00

		>   L2 intestazione 000290

		>>  L1 intestazione 000290-OP1010

		>>> dettaglio record 000290-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000290-OP1010 ore: 1		00

		>   L2 totali 000290 ore: 1		00

		>   L2 intestazione 000300

		>>  L1 intestazione 000300-OP1010

		>>> dettaglio record 000300-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000300-OP1010 ore: 1		00

		>   L2 totali 000300 ore: 1		00

		>   L2 intestazione 000310

		>>  L1 intestazione 000310-OP1010

		>>> dettaglio record 000310-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000310-OP1010 ore: 1		00

		>   L2 totali 000310 ore: 1		00

		>   L2 intestazione 000320

		>>  L1 intestazione 000320-OP2011

		>>> dettaglio record 000320-OP2011-140/1982-01-01 ore: 0		75

		>>> dettaglio record 000320-OP2011-150/1982-01-01 ore: 0		25

		>>  L1 totali 000320-OP2011 ore: 1		00

		>   L2 totali 000320 ore: 1		00

		>   L2 intestazione 000330

		>>  L1 intestazione 000330-OP2012

		>>> dettaglio record 000330-OP2012-140/1982-01-01 ore: 0		25

		>>> dettaglio record 000330-OP2012-160/1982-01-01 ore: 0		75

		>>  L1 totali 000330-OP2012 ore: 1		00

		>   L2 totali 000330 ore: 1		00

		>   L2 intestazione 000340

		>>  L1 intestazione 000340-OP2013

		>>> dettaglio record 000340-OP2013-140/1982-01-01 ore: 0		50

		>>> dettaglio record 000340-OP2013-170/1982-01-01 ore: 0		50

		>>  L1 totali 000340-OP2013 ore: 1		00

		>   L2 totali 000340 ore: 1		00

		LR totali  ore: 54		50




      ****************************************************************
      * ESEMPIO CICLO RPG TRADIZIONALE                               *
      ****************************************************************
      *
      * (c) MarkOneTools - www.markonetools.it - 2023
      *
     H DECEDIT('0,')

     FPROJACT1L IPE  E           K DISK

     DTot1             s                   like(EMPTIME) inz(*zeros)
     DTot2             s                   like(EMPTIME) inz(*zeros)
     DTotG             s                   like(EMPTIME) inz(*zeros)
     DMsg              s            200a   varying

      *------------------- ROTTURE DI LIVELLO ------------------------*
      * rotture più esterne (L9...) a rotture più interne (...L1)
     IPROJA         01
     I                                          EMPNO         L2
     I                                          PROJNO        L1
      * ROTTURE DI LIVELLO
      * elaborate dopo la rottura (in linea il nuovo record)
      * istruzione per l'azzeramento dei totali parziali
      * eventuale decodifica dei codici presenti in testata
      * stampa testate delle rotture di livello
      *------------------------------------------------*
      * ROTTURA LIVELLO L2
      *------------------------------------------------*
     C   L2              do
     C                   clear                   Tot2
     C                   eval      Msg = '>   L2 intestazione ' + EMPNO
     C                   snd-msg   Msg
     C                   enddo
      *------------------------------------------------*
      * ROTTURA LIVELLO L1
      *------------------------------------------------*
     C   L1              do
     C                   clear                   Tot1
     C                   eval      Msg = '>>  L1 intestazione ' + EMPNO + '-' +
     C                                   PROJNO
     C                   snd-msg   Msg
     C                   enddo
      *------------------*
      * DETTAGLIO RECORD
      *------------------*
      * incremento totali del livello 2
     C                   add       EMPTIME       Tot1
     C                   eval      Msg = '>>> dettaglio record ' + EMPNO + '-' +
     C                                   PROJNO + '-' +
     C                                   %char(ACTNO) + '/' + %char(EMSTDATE) +
     C                                   ' ore: ' + %char(EMPTIME)
     C                   snd-msg   Msg

      * ROTTURE DI LIVELLO
      * elaborate alla rottura (in linea il record precedente alla
      * rottura)
      * istruzione per l'incremento del totale generale e dei totali
      * parziali
      * stampa dei record di totale
      * N.B. devono essere le ultime istruzioni del programma
      *------------------*
      * TOTALE LIVELLO L1
      *------------------*
     CL1                 add       Tot1          Tot2
     CL1                 eval      Msg = '>>  L1 totali ' + EMPNO + '-' +
     C                                   PROJNO +
     C                                   ' ore: ' + %char(Tot1)
     CL1                 snd-msg   Msg
      *------------------*
      * TOTALE LIVELLO L2
      *------------------*
     CL2                 add       Tot2          TotG
     CL2                 eval      Msg = '>   L2 totali ' + EMPNO +
     C                                   ' ore: ' + %char(Tot2)
     CL2                 snd-msg   Msg
      *---------------*
      * TOTALE FINALE
      *---------------*
     CLR                 eval      Msg = 'LR totali ' +
     C                                   ' ore: ' + %char(TotG)
     CLR                 snd-msg   Msg
      *
      *****************************************************************
      * subroutine inizializzazione
      *****************************************************************
     C     *inzsr        begsr
     C                   snd-msg   'INIZIO CICLO'
     C                   endsr


		INIZIO CICLO

		>   L2 intestazione 000010

		>>  L1 intestazione 000010-AD3100

		>>> dettaglio record 000010-AD3100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-AD3100 ore: 0		50

		>>  L1 intestazione 000010-MA2100

		>>> dettaglio record 000010-MA2100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-MA2100 ore: 0		50

		>>  L1 intestazione 000010-MA2110

		>>> dettaglio record 000010-MA2110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000010-MA2110 ore: 1		00

		>   L2 totali 000010 ore: 2		00

		>   L2 intestazione 000020

		>>  L1 intestazione 000020-PL2100

		>>> dettaglio record 000020-PL2100-30/1982-01-01 ore: 1		00

		>>  L1 totali 000020-PL2100 ore: 1		00

		>   L2 totali 000020 ore: 1		00

		>   L2 intestazione 000030

		>>  L1 intestazione 000030-IF1000

		>>> dettaglio record 000030-IF1000-10/1982-06-01 ore: 0		50

		>>  L1 totali 000030-IF1000 ore: 0		50

		>>  L1 intestazione 000030-IF2000

		>>> dettaglio record 000030-IF2000-10/1982-01-01 ore: 0		50

		>>  L1 totali 000030-IF2000 ore: 0		50

		>   L2 totali 000030 ore: 1		00

		>   L2 intestazione 000050

		>>  L1 intestazione 000050-OP1000

		>>> dettaglio record 000050-OP1000-10/1982-01-01 ore: 0		25

		>>  L1 totali 000050-OP1000 ore: 0		25

		>>  L1 intestazione 000050-OP2010

		>>> dettaglio record 000050-OP2010-10/1982-01-01 ore: 0		75

		>>  L1 totali 000050-OP2010 ore: 0		75

		>   L2 totali 000050 ore: 1		00

		>   L2 intestazione 000070

		>>  L1 intestazione 000070-AD3110

		>>> dettaglio record 000070-AD3110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000070-AD3110 ore: 1		00

		>   L2 totali 000070 ore: 1		00

		>   L2 intestazione 000090

		>>  L1 intestazione 000090-OP1010

		>>> dettaglio record 000090-OP1010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000090-OP1010 ore: 1		00

		>   L2 totali 000090 ore: 1		00

		>   L2 intestazione 000100

		>>  L1 intestazione 000100-OP2010

		>>> dettaglio record 000100-OP2010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000100-OP2010 ore: 1		00

		>   L2 totali 000100 ore: 1		00

		>   L2 intestazione 000110

		>>  L1 intestazione 000110-MA2100

		>>> dettaglio record 000110-MA2100-20/1982-01-01 ore: 1		00

		>>  L1 totali 000110-MA2100 ore: 1		00

		>   L2 totali 000110 ore: 1		00

		>   L2 intestazione 000130

		>>  L1 intestazione 000130-IF1000

		>>> dettaglio record 000130-IF1000-90/1982-10-01 ore: 1		00

		>>> dettaglio record 000130-IF1000-100/1982-10-01 ore: 0		50

		>>  L1 totali 000130-IF1000 ore: 1		50

		>   L2 totali 000130 ore: 1		50

		>   L2 intestazione 000140

		>>  L1 intestazione 000140-IF1000

		>>> dettaglio record 000140-IF1000-90/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF1000 ore: 0		50

		>>  L1 intestazione 000140-IF2000

		>>> dettaglio record 000140-IF2000-100/1982-01-01 ore: 1		00

		>>> dettaglio record 000140-IF2000-100/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF2000 ore: 2		50

		>   L2 totali 000140 ore: 3		00

		>   L2 intestazione 000150

		>>  L1 intestazione 000150-MA2112

		>>> dettaglio record 000150-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000150-MA2112-180/1982-07-15 ore: 1		00

		>>  L1 totali 000150-MA2112 ore: 2		00

		>   L2 totali 000150 ore: 2		00

		>   L2 intestazione 000160

		>>  L1 intestazione 000160-MA2113

		>>> dettaglio record 000160-MA2113-60/1982-07-15 ore: 1		00

		>>  L1 totali 000160-MA2113 ore: 1		00

		>   L2 totali 000160 ore: 1		00

		>   L2 intestazione 000170

		>>  L1 intestazione 000170-MA2112

		>>> dettaglio record 000170-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000170-MA2112-70/1982-06-01 ore: 1		00

		>>  L1 totali 000170-MA2112 ore: 2		00

		>>  L1 intestazione 000170-MA2113

		>>> dettaglio record 000170-MA2113-80/1982-01-01 ore: 1		00

		>>  L1 totali 000170-MA2113 ore: 1		00

		>   L2 totali 000170 ore: 3		00

		>   L2 intestazione 000180

		>>  L1 intestazione 000180-MA2113

		>>> dettaglio record 000180-MA2113-70/1982-04-01 ore: 1		00

		>>  L1 totali 000180-MA2113 ore: 1		00

		>   L2 totali 000180 ore: 1		00

		>   L2 intestazione 000190

		>>  L1 intestazione 000190-MA2112

		>>> dettaglio record 000190-MA2112-70/1982-01-01 ore: 1		00

		>>> dettaglio record 000190-MA2112-80/1982-10-01 ore: 1		00

		>>  L1 totali 000190-MA2112 ore: 2		00

		>   L2 totali 000190 ore: 2		00

		>   L2 intestazione 000200

		>>  L1 intestazione 000200-MA2111

		>>> dettaglio record 000200-MA2111-50/1982-01-01 ore: 1		00

		>>> dettaglio record 000200-MA2111-60/1982-06-15 ore: 1		00

		>>  L1 totali 000200-MA2111 ore: 2		00

		>   L2 totali 000200 ore: 2		00

		>   L2 intestazione 000210

		>>  L1 intestazione 000210-MA2113

		>>> dettaglio record 000210-MA2113-80/1982-10-01 ore: 0		50

		>>> dettaglio record 000210-MA2113-180/1982-10-01 ore: 0		50

		>>  L1 totali 000210-MA2113 ore: 1		00

		>   L2 totali 000210 ore: 1		00

		>   L2 intestazione 000220

		>>  L1 intestazione 000220-MA2111

		>>> dettaglio record 000220-MA2111-40/1982-01-01 ore: 1		00

		>>  L1 totali 000220-MA2111 ore: 1		00

		>   L2 totali 000220 ore: 1		00

		>   L2 intestazione 000230

		>>  L1 intestazione 000230-AD3111

		>>> dettaglio record 000230-AD3111-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000230-AD3111-60/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-70/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-80/1982-04-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-180/1982-10-15 ore: 0		50

		>>  L1 totali 000230-AD3111 ore: 3		00

		>   L2 totali 000230 ore: 3		00

		>   L2 intestazione 000240

		>>  L1 intestazione 000240-AD3111

		>>> dettaglio record 000240-AD3111-70/1982-02-15 ore: 1		00

		>>> dettaglio record 000240-AD3111-80/1982-09-15 ore: 1		00

		>>  L1 totali 000240-AD3111 ore: 2		00

		>   L2 totali 000240 ore: 2		00

		>   L2 intestazione 000250

		>>  L1 intestazione 000250-AD3112

		>>> dettaglio record 000250-AD3112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-60/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-60/1983-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-70/1982-03-15 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-10-15 ore: 0		50

		>>> dettaglio record 000250-AD3112-180/1982-08-15 ore: 0		50

		>>  L1 totali 000250-AD3112 ore: 5		50

		>   L2 totali 000250 ore: 5		50

		>   L2 intestazione 000260

		>>  L1 intestazione 000260-AD3113

		>>> dettaglio record 000260-AD3113-70/1982-06-15 ore: 0		50

		>>> dettaglio record 000260-AD3113-70/1982-07-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-04-15 ore: 1		00

		>>> dettaglio record 000260-AD3113-180/1982-06-01 ore: 1		00

		>>  L1 totali 000260-AD3113 ore: 5		50

		>   L2 totali 000260 ore: 5		50

		>   L2 intestazione 000270

		>>  L1 intestazione 000270-AD3113

		>>> dettaglio record 000270-AD3113-60/1982-03-01 ore: 0		50

		>>> dettaglio record 000270-AD3113-60/1982-04-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-60/1982-09-01 ore: 0		25

		>>> dettaglio record 000270-AD3113-70/1982-09-01 ore: 0		75

		>>> dettaglio record 000270-AD3113-70/1982-10-15 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-03-01 ore: 0		50

		>>  L1 totali 000270-AD3113 ore: 5		00

		>   L2 totali 000270 ore: 5		00

		>   L2 intestazione 000280

		>>  L1 intestazione 000280-OP1010

		>>> dettaglio record 000280-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000280-OP1010 ore: 1		00

		>   L2 totali 000280 ore: 1		00

		>   L2 intestazione 000290

		>>  L1 intestazione 000290-OP1010

		>>> dettaglio record 000290-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000290-OP1010 ore: 1		00

		>   L2 totali 000290 ore: 1		00

		>   L2 intestazione 000300

		>>  L1 intestazione 000300-OP1010

		>>> dettaglio record 000300-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000300-OP1010 ore: 1		00

		>   L2 totali 000300 ore: 1		00

		>   L2 intestazione 000310

		>>  L1 intestazione 000310-OP1010

		>>> dettaglio record 000310-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000310-OP1010 ore: 1		00

		>   L2 totali 000310 ore: 1		00

		>   L2 intestazione 000320

		>>  L1 intestazione 000320-OP2011

		>>> dettaglio record 000320-OP2011-140/1982-01-01 ore: 0		75

		>>> dettaglio record 000320-OP2011-150/1982-01-01 ore: 0		25

		>>  L1 totali 000320-OP2011 ore: 1		00

		>   L2 totali 000320 ore: 1		00

		>   L2 intestazione 000330

		>>  L1 intestazione 000330-OP2012

		>>> dettaglio record 000330-OP2012-140/1982-01-01 ore: 0		25

		>>> dettaglio record 000330-OP2012-160/1982-01-01 ore: 0		75

		>>  L1 totali 000330-OP2012 ore: 1		00

		>   L2 totali 000330 ore: 1		00

		>   L2 intestazione 000340

		>>  L1 intestazione 000340-OP2013

		>>> dettaglio record 000340-OP2013-140/1982-01-01 ore: 0		50

		>>> dettaglio record 000340-OP2013-170/1982-01-01 ore: 0		50

		>>  L1 totali 000340-OP2013 ore: 1		00

		>   L2 totali 000340 ore: 1		00

		LR totali  ore: 54		50




**free

// SIMULAZIONE CICLO RPG CON EMBEDDED SQL
// (c) MarkOneTools - www.markonetools.it - 2023

// versione con gestione rotture di livello equivalenti al ciclo tradizionale
// funzioni SQL: lag e lead

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy: *nounref)
        expropts(*resdecpos)
        extbinint(*yes);

dcl-c sqsc_ok 0;
dcl-c sqsc_norek 100;

dcl-ds EMPPROJACT ext template end-ds;
dcl-ds oAct qualified inz;
  L2h ind;
  L2t ind;
  L1h ind;
  L1t ind;
  EMPNO    like(EMPNO);
  PROJNO   like(PROJNO);
  ACTNO    like(ACTNO);
  EMSTDATE like(EMSTDATE);
  EMPTIME  like(EMPTIME);
end-ds;

dcl-s Tot1 like(EMPTIME);
dcl-s Tot2 like(EMPTIME);
dcl-s TotG like(EMPTIME);
dcl-s Msg varchar(200);

exec sql
  declare cAct cursor for
    select case when lag(EMPNO) over(order by order of EMPPROJACT)
                      is distinct from EMPNO
                then '1' else '0' end "L2 int.",
           case when lead(EMPNO) over(order by order of EMPPROJACT)
                      is distinct from EMPNO
                then '1' else '0' end "L2 tot.",
           case when lag(PROJNO) over(partition by EMPNO order by order of EMPPROJACT)
                      is distinct from PROJNO
                then '1' else '0' end "L1 int.",
           case when lead(PROJNO) over(partition by EMPNO order by order of EMPPROJACT)
                      is distinct from PROJNO
                then '1' else '0' end "L1 tot.",
           EMPNO, PROJNO, ACTNO, EMSTDATE, EMPTIME
      from EMPPROJACT
      order by EMPNO, PROJNO, ACTNO;

// INIZIO DEL CICLO
snd-msg 'INIZIO CICLO';

exec sql
  open cAct;

dow *on;
  exec sql
    fetch next
    from cAct
    into :oAct;

  if sqlcode = sqsc_norek;
    leave;
  endif;

  //--- ROTTURE LIVELLO INTESTAZIONE
  // ROTTURA LIVELLO L2
  if oAct.L2h;
    clear Tot2;
    Msg = '>   L2 intestazione ' + oAct.EMPNO;
    snd-msg Msg;
  endif;
  // ROTTURA LIVELLO L1
  if oAct.L1h;
    clear Tot1;
    Msg = '>>  L1 intestazione ' + oAct.EMPNO + '-' + oAct.PROJNO;
    snd-msg Msg;
  endif;

  //--- DETTAGLIO RECORD
  Tot1 += oAct.EMPTIME;
  Msg = '>>> dettaglio record ' + oAct.EMPNO + '-' + oAct.PROJNO + '-' + %char(oAct.ACTNO) +
        '/' + %char(oAct.EMSTDATE) +
        ' ore: ' + %char(oAct.EMPTIME);
  snd-msg Msg;

  //--- ROTTURE LIVELLO TOTALI
  // TOTALI LIVELLO L1
  if oAct.L1t;
    Tot2 += Tot1;
    Msg = '>>  L1 totali ' + oAct.EMPNO + '-' + oAct.PROJNO + ' ore: ' + %char(Tot1);
    snd-msg Msg;
  endif;
  // TOTALI LIVELLO L2
  if oAct.L2t;
    TotG += Tot2;
    Msg = '>   L2 totali ' + oAct.EMPNO + ' ore: ' + %char(Tot2);
    snd-msg Msg;
  endif;
enddo;

// TOTALI FINALI
Msg = 'LR totali ' + ' ore: ' + %char(TotG);
snd-msg Msg;

exec sql
  close cAct;

*inlr = *on;
return;


		INIZIO CICLO

		>   L2 intestazione 000010

		>>  L1 intestazione 000010-AD3100

		>>> dettaglio record 000010-AD3100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-AD3100 ore: 0		50

		>>  L1 intestazione 000010-MA2100

		>>> dettaglio record 000010-MA2100-10/1982-01-01 ore: 0		50

		>>  L1 totali 000010-MA2100 ore: 0		50

		>>  L1 intestazione 000010-MA2110

		>>> dettaglio record 000010-MA2110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000010-MA2110 ore: 1		00

		>   L2 totali 000010 ore: 2		00

		>   L2 intestazione 000020

		>>  L1 intestazione 000020-PL2100

		>>> dettaglio record 000020-PL2100-30/1982-01-01 ore: 1		00

		>>  L1 totali 000020-PL2100 ore: 1		00

		>   L2 totali 000020 ore: 1		00

		>   L2 intestazione 000030

		>>  L1 intestazione 000030-IF1000

		>>> dettaglio record 000030-IF1000-10/1982-06-01 ore: 0		50

		>>  L1 totali 000030-IF1000 ore: 0		50

		>>  L1 intestazione 000030-IF2000

		>>> dettaglio record 000030-IF2000-10/1982-01-01 ore: 0		50

		>>  L1 totali 000030-IF2000 ore: 0		50

		>   L2 totali 000030 ore: 1		00

		>   L2 intestazione 000050

		>>  L1 intestazione 000050-OP1000

		>>> dettaglio record 000050-OP1000-10/1982-01-01 ore: 0		25

		>>  L1 totali 000050-OP1000 ore: 0		25

		>>  L1 intestazione 000050-OP2010

		>>> dettaglio record 000050-OP2010-10/1982-01-01 ore: 0		75

		>>  L1 totali 000050-OP2010 ore: 0		75

		>   L2 totali 000050 ore: 1		00

		>   L2 intestazione 000070

		>>  L1 intestazione 000070-AD3110

		>>> dettaglio record 000070-AD3110-10/1982-01-01 ore: 1		00

		>>  L1 totali 000070-AD3110 ore: 1		00

		>   L2 totali 000070 ore: 1		00

		>   L2 intestazione 000090

		>>  L1 intestazione 000090-OP1010

		>>> dettaglio record 000090-OP1010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000090-OP1010 ore: 1		00

		>   L2 totali 000090 ore: 1		00

		>   L2 intestazione 000100

		>>  L1 intestazione 000100-OP2010

		>>> dettaglio record 000100-OP2010-10/1982-01-01 ore: 1		00

		>>  L1 totali 000100-OP2010 ore: 1		00

		>   L2 totali 000100 ore: 1		00

		>   L2 intestazione 000110

		>>  L1 intestazione 000110-MA2100

		>>> dettaglio record 000110-MA2100-20/1982-01-01 ore: 1		00

		>>  L1 totali 000110-MA2100 ore: 1		00

		>   L2 totali 000110 ore: 1		00

		>   L2 intestazione 000130

		>>  L1 intestazione 000130-IF1000

		>>> dettaglio record 000130-IF1000-90/1982-10-01 ore: 1		00

		>>> dettaglio record 000130-IF1000-100/1982-10-01 ore: 0		50

		>>  L1 totali 000130-IF1000 ore: 1		50

		>   L2 totali 000130 ore: 1		50

		>   L2 intestazione 000140

		>>  L1 intestazione 000140-IF1000

		>>> dettaglio record 000140-IF1000-90/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF1000 ore: 0		50

		>>  L1 intestazione 000140-IF2000

		>>> dettaglio record 000140-IF2000-100/1982-01-01 ore: 1		00

		>>> dettaglio record 000140-IF2000-100/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-03-01 ore: 0		50

		>>> dettaglio record 000140-IF2000-110/1982-10-01 ore: 0		50

		>>  L1 totali 000140-IF2000 ore: 2		50

		>   L2 totali 000140 ore: 3		00

		>   L2 intestazione 000150

		>>  L1 intestazione 000150-MA2112

		>>> dettaglio record 000150-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000150-MA2112-180/1982-07-15 ore: 1		00

		>>  L1 totali 000150-MA2112 ore: 2		00

		>   L2 totali 000150 ore: 2		00

		>   L2 intestazione 000160

		>>  L1 intestazione 000160-MA2113

		>>> dettaglio record 000160-MA2113-60/1982-07-15 ore: 1		00

		>>  L1 totali 000160-MA2113 ore: 1		00

		>   L2 totali 000160 ore: 1		00

		>   L2 intestazione 000170

		>>  L1 intestazione 000170-MA2112

		>>> dettaglio record 000170-MA2112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000170-MA2112-70/1982-06-01 ore: 1		00

		>>  L1 totali 000170-MA2112 ore: 2		00

		>>  L1 intestazione 000170-MA2113

		>>> dettaglio record 000170-MA2113-80/1982-01-01 ore: 1		00

		>>  L1 totali 000170-MA2113 ore: 1		00

		>   L2 totali 000170 ore: 3		00

		>   L2 intestazione 000180

		>>  L1 intestazione 000180-MA2113

		>>> dettaglio record 000180-MA2113-70/1982-04-01 ore: 1		00

		>>  L1 totali 000180-MA2113 ore: 1		00

		>   L2 totali 000180 ore: 1		00

		>   L2 intestazione 000190

		>>  L1 intestazione 000190-MA2112

		>>> dettaglio record 000190-MA2112-70/1982-01-01 ore: 1		00

		>>> dettaglio record 000190-MA2112-80/1982-10-01 ore: 1		00

		>>  L1 totali 000190-MA2112 ore: 2		00

		>   L2 totali 000190 ore: 2		00

		>   L2 intestazione 000200

		>>  L1 intestazione 000200-MA2111

		>>> dettaglio record 000200-MA2111-50/1982-01-01 ore: 1		00

		>>> dettaglio record 000200-MA2111-60/1982-06-15 ore: 1		00

		>>  L1 totali 000200-MA2111 ore: 2		00

		>   L2 totali 000200 ore: 2		00

		>   L2 intestazione 000210

		>>  L1 intestazione 000210-MA2113

		>>> dettaglio record 000210-MA2113-80/1982-10-01 ore: 0		50

		>>> dettaglio record 000210-MA2113-180/1982-10-01 ore: 0		50

		>>  L1 totali 000210-MA2113 ore: 1		00

		>   L2 totali 000210 ore: 1		00

		>   L2 intestazione 000220

		>>  L1 intestazione 000220-MA2111

		>>> dettaglio record 000220-MA2111-40/1982-01-01 ore: 1		00

		>>  L1 totali 000220-MA2111 ore: 1		00

		>   L2 totali 000220 ore: 1		00

		>   L2 intestazione 000230

		>>  L1 intestazione 000230-AD3111

		>>> dettaglio record 000230-AD3111-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000230-AD3111-60/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-70/1982-03-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-80/1982-04-15 ore: 0		50

		>>> dettaglio record 000230-AD3111-180/1982-10-15 ore: 0		50

		>>  L1 totali 000230-AD3111 ore: 3		00

		>   L2 totali 000230 ore: 3		00

		>   L2 intestazione 000240

		>>  L1 intestazione 000240-AD3111

		>>> dettaglio record 000240-AD3111-70/1982-02-15 ore: 1		00

		>>> dettaglio record 000240-AD3111-80/1982-09-15 ore: 1		00

		>>  L1 totali 000240-AD3111 ore: 2		00

		>   L2 totali 000240 ore: 2		00

		>   L2 intestazione 000250

		>>  L1 intestazione 000250-AD3112

		>>> dettaglio record 000250-AD3112-60/1982-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-60/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-60/1983-01-01 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-02-01 ore: 0		50

		>>> dettaglio record 000250-AD3112-70/1982-03-15 ore: 1		00

		>>> dettaglio record 000250-AD3112-70/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-08-15 ore: 0		25

		>>> dettaglio record 000250-AD3112-80/1982-10-15 ore: 0		50

		>>> dettaglio record 000250-AD3112-180/1982-08-15 ore: 0		50

		>>  L1 totali 000250-AD3112 ore: 5		50

		>   L2 totali 000250 ore: 5		50

		>   L2 intestazione 000260

		>>  L1 intestazione 000260-AD3113

		>>> dettaglio record 000260-AD3113-70/1982-06-15 ore: 0		50

		>>> dettaglio record 000260-AD3113-70/1982-07-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000260-AD3113-80/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-03-01 ore: 0		50

		>>> dettaglio record 000260-AD3113-180/1982-04-15 ore: 1		00

		>>> dettaglio record 000260-AD3113-180/1982-06-01 ore: 1		00

		>>  L1 totali 000260-AD3113 ore: 5		50

		>   L2 totali 000260 ore: 5		50

		>   L2 intestazione 000270

		>>  L1 intestazione 000270-AD3113

		>>> dettaglio record 000270-AD3113-60/1982-03-01 ore: 0		50

		>>> dettaglio record 000270-AD3113-60/1982-04-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-60/1982-09-01 ore: 0		25

		>>> dettaglio record 000270-AD3113-70/1982-09-01 ore: 0		75

		>>> dettaglio record 000270-AD3113-70/1982-10-15 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-01-01 ore: 1		00

		>>> dettaglio record 000270-AD3113-80/1982-03-01 ore: 0		50

		>>  L1 totali 000270-AD3113 ore: 5		00

		>   L2 totali 000270 ore: 5		00

		>   L2 intestazione 000280

		>>  L1 intestazione 000280-OP1010

		>>> dettaglio record 000280-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000280-OP1010 ore: 1		00

		>   L2 totali 000280 ore: 1		00

		>   L2 intestazione 000290

		>>  L1 intestazione 000290-OP1010

		>>> dettaglio record 000290-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000290-OP1010 ore: 1		00

		>   L2 totali 000290 ore: 1		00

		>   L2 intestazione 000300

		>>  L1 intestazione 000300-OP1010

		>>> dettaglio record 000300-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000300-OP1010 ore: 1		00

		>   L2 totali 000300 ore: 1		00

		>   L2 intestazione 000310

		>>  L1 intestazione 000310-OP1010

		>>> dettaglio record 000310-OP1010-130/1982-01-01 ore: 1		00

		>>  L1 totali 000310-OP1010 ore: 1		00

		>   L2 totali 000310 ore: 1		00

		>   L2 intestazione 000320

		>>  L1 intestazione 000320-OP2011

		>>> dettaglio record 000320-OP2011-140/1982-01-01 ore: 0		75

		>>> dettaglio record 000320-OP2011-150/1982-01-01 ore: 0		25

		>>  L1 totali 000320-OP2011 ore: 1		00

		>   L2 totali 000320 ore: 1		00

		>   L2 intestazione 000330

		>>  L1 intestazione 000330-OP2012

		>>> dettaglio record 000330-OP2012-140/1982-01-01 ore: 0		25

		>>> dettaglio record 000330-OP2012-160/1982-01-01 ore: 0		75

		>>  L1 totali 000330-OP2012 ore: 1		00

		>   L2 totali 000330 ore: 1		00

		>   L2 intestazione 000340

		>>  L1 intestazione 000340-OP2013

		>>> dettaglio record 000340-OP2013-140/1982-01-01 ore: 0		50

		>>> dettaglio record 000340-OP2013-170/1982-01-01 ore: 0		50

		>>  L1 totali 000340-OP2013 ore: 1		00

		>   L2 totali 000340 ore: 1		00

		LR totali  ore: 54		50




**free

// SIMULAZIONE CICLO RPG CON EMBEDDED SQL
// (c) MarkOneTools - www.markonetools.it - 2023

// versione con gestione rotture di livello e totalizzazioni con SQL
// supergruppi SQL e funzione SQL grouping

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy: *nounref)
        expropts(*resdecpos)
        extbinint(*yes);

dcl-c sqsc_ok 0;
dcl-c sqsc_norek 100;

dcl-ds EMPPROJACT ext template end-ds;
dcl-ds oAct qualified inz;
  LRt ind;
  L2h ind;
  L2t ind;
  L1h ind;
  L1t ind;
  EMPNO       like(EMPNO);
  PROJNO      like(PROJNO);
  ACTNO       like(ACTNO);
  EMSTDATE    like(EMSTDATE);
  TOT_EMPTIME like(EMPTIME);
end-ds;
dcl-s iAct int(5) dim(10);

dcl-s Msg varchar(200);

exec sql
  declare cAct cursor for
    with
     TOT as
     (select EMPNO, PROJNO,
           sum(EMPTIME) as EMPTIME
      from EMPPROJACT
      group by EMPNO, PROJNO with rollup),
     DETT as
     (select EMPNO, PROJNO, ACTNO, EMSTDATE, EMPTIME
      from EMPPROJACT),
     RES as
     (select EMPNO, PROJNO,
             cast(null as integer) as ACTNO, cast(null as date) as EMSTDATE, EMPTIME
        from TOT
      union all
      select *
        from DETT)
     select case when EMPNO is null and PROJNO is null then '1' else '0' end "LR tot.",
            case when lag(EMPNO)  over(order by order of RES)
                       is distinct from EMPNO and EMPNO is not null
                 then '1' else '0' end "L2 int.",
            case when lead(EMPNO) over(order by order of RES)
                       is distinct from EMPNO
                 then '1' else '0' end "L2 tot.",
            case when lag(PROJNO)  over(partition by EMPNO order by order of RES)
                       is distinct from PROJNO and PROJNO is not null
                 then '1' else '0' end "L1 int.",
            case when lead(PROJNO) over(partition by EMPNO order by order of RES)
                       is distinct from PROJNO
                 then '1' else '0' end "L1 tot.",
            RES.*
       from RES
       order by EMPNO, PROJNO, ACTNO;

// INIZIO DEL CICLO
snd-msg 'INIZIO CICLO';

exec sql
  open cAct;

dow *on;
  exec sql
    fetch next
    from cAct
    into :oAct :iAct;

  if sqlcode = sqsc_norek;
    leave;
  endif;

  //--- ROTTURE LIVELLO INTESTAZIONE
  // ROTTURA LIVELLO L2
  if oAct.L2h;
    Msg = '>   L2 intestazione ' + oAct.EMPNO;
    snd-msg Msg;
  endif;
  // ROTTURA LIVELLO L1
  if oAct.L1h;
    Msg = '>>  L1 intestazione ' + oAct.EMPNO + '-' + oAct.PROJNO;
    snd-msg Msg;
  endif;


  //--- ROTTURE LIVELLO TOTALI
  select;
    // TOTALI FINALI
    when oAct.LRt;
      Msg = 'LR totali ' + ' ore: ' + %char(oAct.TOT_EMPTIME);
      snd-msg Msg;
    // TOTALI LIVELLO L2
    when oAct.L2t;
      Msg = '>   L2 totali ' + oAct.EMPNO + ' ore: ' + %char(oAct.TOT_EMPTIME);
      snd-msg Msg;
    // TOTALI LIVELLO L1
    when oAct.L1t;
      Msg = '>>  L1 totali ' + oAct.EMPNO + '-' + oAct.PROJNO + ' ore: ' + %char(oAct.TOT_EMPTIME);
      snd-msg Msg;
    //--- DETTAGLIO RECORD
    other;
      Msg = '>>> dettaglio record ' + oAct.EMPNO + '-' + oAct.PROJNO + '-' + %char(oAct.ACTNO) +
            '/' + %char(oAct.EMSTDATE) +
            ' ore: ' + %char(oAct.TOT_EMPTIME);
      snd-msg Msg;
  endsl;
enddo;

exec sql
  close cAct;

*inlr = *on;
return;


-- tabella employee project activities: indice per report
drop index if exists EMPPROJACT_1L;
create index EMPPROJACT_1L for system name PROJACT1L on EMPPROJACT (EMPNO, PROJNO, ACTNO)
  rcdfmt PROJA add all columns;
label on index EMPPROJACT_1L is 'RI per impiegato/progetto/attività';

Impiegato 000330

  Progetto OP2012

            140          01-01-1982     0,25

            160          01-01-1982     0,75

  Tot. Progetto OP2012                  1,00

Tot.impiegato 000330                    1,00



Impiegato 000340

  Progetto OP2013

            140          01-01-1982     0,50

            170          01-01-1982     0,50

  Tot. Progetto OP2013                  1,00

Tot.impiegato 000340                    1,00



TOT.GENERALE                           54,50

